
A monitor synthesis algorithm from linear temporal logic (LTL) safety for-
mulae of the form �ϕ where ϕ is a past time LTL formula was presented in [3].
The generated monitors implemented the recursive semantics of past-time LTL
using a dynamic programming technique, and needed O(|ϕ|) time to process
each new event and O(|ϕ|) total space. Some compiler-like optimizations of the
generated monitors were also proposed in [3], which would further reduce the
required space. It is not clear how much the required space could be reduced by
applying those optimizations.

We here show how to generate using a divide-and-conquer technique directly
monitors that need O(k) space and still O(|ϕ|) time, where k is the number of
temporal operators in ϕ.

1 The Monitor Synthesis Algorithm

For simplicity, we assume only two past operators, namely ◦· (previously) and
S (since). Let us first note that one cannot asymptotically reduce the space
requirements below Ω(k), where k is the number of temporal operators appearing
in the formula to monitor ϕ. Indeed, one can take ϕ = (#1 → t1)∧ · · · ∧ (#k →
tk), where for each 1 ≤ i ≤ k, #i is some event and ti is some temporal formula
containing precisely one past temporal operator, i.e., a ◦· or a S . Any monitor
for ϕ must directly or indirectly store the status of each ti at every event, to be
able to react accordingly in case the next event is some #i. Assuming that the
events #i are distinct and that the formulae ti are unrelated, then the monitor
needs to distinguish among 2k possible states, so it needs Ω(k) space.

In what follows, we assume the usual recursive semantics of LTL, also pre-
sented below, restricted to safety formulae of the form �ϕ, where ϕ is a past-time
LTL. We adopt the simplifying assumption that the empty trace invalidates any
atomic proposition and any past temporal operator; as argued in [3], this may
not always be the best choice, but other semantic variations regarding the empty
trace present no difficulties for monitoring.

Definition 1. (adapted from [4]) LTL formulae of the form �ϕ (read “always
ϕ”), where ϕ is a past-time LTL formula, are called LTL safety formulae; we
may call them just safety formulae when LTL is understood from the context.
An infinite trace u ∈ Σω satisfies �ϕ, written u |= �ϕ, iff each w ∈ prefixes(u)



satisfies the past-time LTL formula ϕ, written also w |= ϕ and defined inductively
as follows:

w |= true is always true,
ws |= a iff a(s) holds,
w |= ¬ϕ iff w �|= ϕ,
w |= ϕ1 ∧ ϕ2 iff w |= ϕ1 and w |= ϕ2,
ws |= ◦·ϕ iff w |= ϕ,
ws |= ϕ1 S ϕ2 iff ws |= ϕ2 or ws |= Fϕ and w |= ϕ1 S ϕ2

ε |= ϕ is false otherwise

Given safety formula �ϕ, we let L(�ϕ) ⊆ Σω be the set {u ∈ Σω | u |= �ϕ}.
Let us next investigate the problem of monitoring safety properties P ex-

pressed as languages of safety formulae, that is, P = L(�ϕ) for some past-time
LTL formula ϕ. Because of the recursive nature of the satisfaction relation, a
first important observation is that the generated monitor only needs to store
information regarding the status of temporal operators from the previous state.
More precisely, the monitor needs one bit per temporal operator, keeping the
satisfaction status of the subformula corresponding to that temporal operator;
when a new state is received, the satisfaction status of the subformula is recal-
culated according to the recursive semantics above and then the bit is updated.
The order in which the temporal operators are processed when a new state is
received is important: the nested operators must be processed first.

We next present the actual monitor synthesis algorithm at a high-level. We
refrain from giving detailed pseudocode as we did in [3], because different ap-
plications may choose different implementation paradigms. For example, we are
currently using rewriting techniques to implement the monitor synthesis algo-
rithms in MOP [1]; Section 1.1 shows our complete Maude rewriting implemen-
tation of the subsequent monitor synthesis algorithm.

Step 1 Let ϕ1, ..., ϕk be the k subformulae of ϕ corresponding to temporal
operators, such that, if ϕi is a subformula of ϕj , then i < j; this can be
easily achieved by a DFS traversal of ϕ.

Step 2 Let bit[1..k] be a vector of k bits initialized to 0 (or false); bit[i] will
store information related to ϕi from the previous state:
– if ϕi = ◦·ψ then bit[i] says if ψ was satisfied at the previous state;
– if ϕi = ψ S ψ′ then bit[i] says if ϕi was satisfied at the previous step.

Step 3 Let bit′[1..k] be another vector of k bits; this will be used to store tem-
porary results, which will be moved eventually into the vector bit[1..k].

Step 4 Generate a loop that executes whenever a new state s is available; the
body of the loop executes the following code:
Step 4.1 For each i from 1 to k execute a bit assignment as follows, where

for a subformula ψ of ϕ, ψ is the boolean expression replacing in ψ
each non-nested temporal subformula ϕj by bit[j] if ϕj is a “previously”
formula or by bit′[j] if ϕj is a “since” formula, and each remaining atomic
proposition a by its satisfaction in the current state, a(s):

– if ϕi = ◦·ψ then generate the assignment bit′[i] := ψ
– if ϕi = ψ S ψ′ then generate the assignment bit′[i] := ψ′ ∨ ψ ∧ bit[i]

Step 4.2 Generate the conditional: if ϕ is false then error (formula violated)
Step 4.3 Generate code to move the contents of bit′[1..k] into bit[1..k].

Note that the generated monitors are well-defined, because each time a ψ boolean
expression is generated, all the bits in bit′[1..k] that are needed are already
calculated. One can also perform boolean simplifications when calculating ψ
to reduce runtime overhead even further. For example, in our implementation
that also generated the code below (see Section 1.1), we used the simplification
¬¬ψ = ψ. To illustrate the monitor generation algorithm above, let us consider
the past time LTL formula: ϕ = ¬(a ∧ ¬(◦· b ∧ (c S (d ∧ (¬e S f))))). Step 1
produces the following enumeration of ϕ’s subformulae: ϕ1 = ◦· b, ϕ2 = ¬e S f ,
and ϕ3 = c S (d ∧ (¬e S f)). The other steps eventually generate the code:

bit[1..3] := false; // three global bits
foreach new state s do {

// first update the bits in a consistent order
bit′[1] := b(s);
bit′[2] := f(s) ∨ (¬e(s) ∧ bit[2]);
bit′[3] := d(s) ∧ bit′[2]) ∨ (c(s) ∧ bit[3]);
// then check whether the formula is violated
if a(s) ∧ ¬(bit[1] ∧ bit′[3]) then Error;
// finally, update the state of the monitor
bit[1..3] := bit′[1..3]

}

It is easy to see that for any past LTL formula ϕ of k temporal operators, the
state of the generated monitor is encoded on k bits, namely the vector bit[1..k].
The runtime of the generated monitor is still O(|ϕ|), because each temporal
operator in ϕ results in an assignment and a read operation in the monitor,
while each boolean operator in ϕ is “executed” by the monitor.

1.1 A Maude Implementation of the Monitor Synthesizer

We here show a term rewriting implementation of the algorithm above, using
the Maude system [2]. Implementations in other languages are obviously also
possible; however, rewriting proved to be an elegant means to generate monitors
from logical formulae in several other contexts, and so seems to be here. In what
follows we show the complete Maude code that takes as input a formula, parses
it, generates the monitor, and then pretty prints it. We use the K technique
here [5], which is a rewriting-based language and/or logic definitional technique;
to use K, one needs to first upload the generic, i.e., application-independent,
module discussed at the end of this section.

Atomic Predicates We start by defining the atomic state predicates that one
can use in formulae. These can be either identifiers (of the form ’a, ’abc, ’a123,
etc.; these are provided by the Maude builtin module QID):

fmod PREDICATE is

--- atomic predicates can be quoted identifiers

protecting QID .

sort Predicate .

subsort Qid < Predicate .

endfm

Syntax of Formulae Let us next define the syntax of formulae. We here use
Maude’s mixfix notation for defining syntax as algebraic operators, where un-
derscores stay for arguments. Also, note that operators are assigned precedences
(declared as operator attributes), to relive the user from writing parentheses (the
lower the precedence the tighter the binding):

fmod SYNTAX is

protecting PREDICATE .

sort Formula .

subsort Predicate < Formula .

op !_ : Formula -> Formula [prec 20] .

op _/\_ : Formula Formula -> Formula [prec 23] .

op O_ : Formula -> Formula [prec 21] .

op _S_ : Formula Formula -> Formula [prec 22] .

endfm

Target Language We are done with the input language. Let us now define
the output language. We need a very simple language for implementing the gen-
erated monitors, namely one with limited assignment, conditional and looping.
The generated code, as well as the target language, play no role in this paper;
one is expected to change the language below to one’s desired target language
(Java, C, C#, assembler, etc.). Our chosen language below has bits, expressions,
statements and code. Bits are also expressions; code is a list of statements com-
posed sequentially using “;” or just concatenation. The syntax below is also
making use of precedence attributes. The format attributes are useful solely for
pretty-printing reasons (see Maude’s manual [2] for details on formatting):

fmod CODE is

--- syntax for the generated code

protecting PREDICATE + INT + STRING .

sorts Bit Exp Statement Code .

subsorts Bit < Exp .

subsort Statement < Code .

ops (bit[_]) (bit’[_]) : Nat -> Bit .

ops (bit[1 .. _]) (bit’[1 .. _]) : Int -> Bit .

op _(s) : Predicate -> Exp [prec 0] .

ops true false : -> Exp .

op !_ : Exp -> Exp [prec 20] .

op _/\_ : Exp Exp -> Exp [prec 23] .

op _\/_ : Exp Exp -> Exp [prec 24] .

op _:=_ : Exp Exp -> Statement [prec 27 format(ni d d d)] .

op if_then_ : Exp Statement -> Statement

[format(ni d d ++ --) prec 30] .

op foreach new state s do _ : Code -> Statement

[format(n d d d d s++ --n)] .

op Error : -> Statement [format(ni d)] .

op //_ : String -> Statement [format(ni d d)] .

op nil : -> Code .

op _;_ : Code Code -> Code [assoc id: nil prec 40] .

op __ : Code Code -> Code [assoc id: nil prec 40] .

op {_} : Code -> Statement [format(d d --ni ++)] .

--- code simplification rules

var B : Exp .

eq ! ! B = B .

endfm

The following module defines the actual monitor synthesis algorithm. We use
the K definitional technique here, because it yields a very compact implemen-
tation. K is centered on the basic intuition of computation; computations are
encoded as first-order data-structures that “evolve”, via rewriting, to results.
Computations are sequentialized using the list constructor “ -> ”; thus, if K and
K’ are computations, then K -> K’ is the computation consisting of K followed
by K’. Computations may eventually yield results; for example, K -> K’ may
rewrite (in context) to R -> K’, meaning that R is the result that K reduces to.
An important feature of K is that one can schedule lists of tasks for reduction; for
example, [K1,K2,K3] -> K may eventually reduce to [R1,R2,R3] -> K, where
R1, R2, and R3 are the results that K1, K2, and K3 reduce to, in this order. To
use K, one needs to import the module K discussed at the and of this section.
The equations of the module K (three in total) are all about reducing a list of
computations to a list of results, supposing that one knows how to reduce one
computation to one result.

K is a definitional framework that is generic in computations and results.
More precisely, it provides sorts KComputation and KResult, and expects its
user to define the desired computations and results, as well as rules to reduce
a computation to a result. Computations typically can be reduced to results
only in context; to facilitate this, K provides a sort KConfiguration, which is
also supposed to be populated accordingly. The sort KConfiguration is a multi-
set sort over a sort KConfigurationItem, where the multi-set constructor is just
concatenation; also, the sort KComputation is a list sort over KComputationItem,
where the list constructor is -> . To make use of K, one needs to first define con-
structors for the sorts KConfigurationItem, KComputationItem and KResult,
and then to define how each computation item reduces to a result.

In our case, the computations are the formulae or subformulae that still need
to be processed, and the results are the corresponding boolean expressions that

need to be checked in the current (generated code) context to see whether the
formula has been violated or not. We define the following additional constructors:
we add four constructors for configurations, namely “k” that wraps the current
computation, “code” that wraps the current generated code, and “nextBit”
that wraps the next available bit; we add one main constructor for computations,
“form”, that wraps a formula, and one constant computation item per operator
in the input language (the later is needed to know how to combine back the
results of the corresponding subexpressions; finally, we add one constructor for
results, “exp”, that wraps a boolean expression.

The formula is processed in a depth-first-order, following a divide-and-conquer
philosophy. Each subformula is decomposed into a list of computation sub-
tasks consisting of its subformulae, then the corresponding results are composed
back into a result corresponding to the original subformula. Recall that equa-
tions/rules apply wherever they match, not only at the top. Let us only discuss
the two equations defining the “since” ( S ), the last two in the module below.
The first one is straightforward: it decomposes the task of processing F1 S F2
to the subtasks of processing F1 and F2; the computation item S is placed in the
computation structure to prepare the terrain for the next equation. The next
equation applies after F1 and F2 have been processed, say to expressions B1 and
B2, respectively; if C is the code generated so far and if I+1 is the next bit avail-
able, then the boolean expression corresponding to the current since formula is
indeed bit’(I+1), provided that one adds the corresponding code capturing the
recursive semantics of since to the generated code.

fmod MONITOR-GENERATION is

protecting K + SYNTAX + CODE .

op k : KComputation -> KConfigurationItem .

op code : Code -> KConfigurationItem .

op nextBit : Nat -> KConfigurationItem .

op process : Formula -> KConfiguration .

op form : Formula -> KComputationItem .

op exp : Exp -> KResult .

ops ! /\ O S : -> KComputationItem .

var P : Predicate . vars F F1 F2 : Formula . var C : Code .

var I : Nat . vars B B1 B2 : Exp . var K : KComputation .

eq process(F) = k(form(F)) code(nil) nextBit(0) .

eq k(form(P) -> K) = k(exp(P(s)) -> K) .

eq form(! F) = form(F) -> ! .

eq exp(B) -> ! = exp(! B) .

eq form(F1 /\ F2) = [form(F1),form(F2)] -> /\ .

eq [exp(B1),exp(B2)] -> /\ = exp(B1 /\ B2) .

eq form(O F) = form(F) -> O .

eq k(exp(B) -> O -> K) code(C) nextBit(I)

= k(exp(bit[I + 1]) -> K) code(C ; bit’[I + 1] := B)

nextBit(I + 1) .

eq form(F1 S F2) = [form(F1), form(F2)] -> S .

eq k([exp(B1),exp(B2)] -> S -> K) code(C) nextBit(I)

= k(exp(bit’[I + 1]) -> K)

code(C ; bit’[I + 1] := B2 \/ B1 /\ bit[I + 1]) nextBit(I + 1) .

endfm

Putting It All together The following module plugs the code generated above
into the general pattern:

fmod PRETTY-PRINT is

protecting MONITOR-GENERATION .

sort Monitor .

op genMonitor : Formula -> Code .

op makeMonitor : KConfiguration -> Code .

var F : Formula . var B : Exp . var C : Code . vars N M : Nat .

eq genMonitor(F) = makeMonitor(process(F)) .

eq makeMonitor(k(exp(B)) code(C) nextBit(N))

= bit[1 .. N] := false ;

foreach new state s do {

// "first update the bits in a consistent order"

C ;

// "then check whether the formula is violated"

if !(B) then Error ;

// "finally, update the state of the monitor"

bit[1 .. N] := bit’[1 .. N]

} .

endfm

Our implementation of the monitor synthesizer is now complete. To use it,
one can ask Maude reduce terms of the form genMonitor(F), where F is the
formula that one wants to generate into a monitor. For example:

reduce genMonitor(

!(’a /\ !(O ’b /\ ’c S (’d /\ (! ’e S ’f))))

) .

For the formula above, Maude will give the expected answer, pretty printed
as follows:

\||||||||||||||||||/

--- Welcome to Maude ---

/||||||||||||||||||\

Maude 2.2 built: Mar 15 2006 16:37:22

Copyright 1997-2005 SRI International

Sat Jan 27 12:01:20 2007

Maude> in p

==========================================

fmod K

==========================================

fmod PREDICATE

==========================================

fmod SYNTAX

==========================================

fmod CODE

==========================================

fmod MONITOR-GENERATION

==========================================

fmod PRETTY-PRINT

==========================================

reduce in PRETTY-PRINT :

genMonitor(! (’a /\ ! (O ’b /\ ’c S (’d /\ ! ’e S ’f)))) .

rewrites: 46 in -93406740ms cpu (1ms real) (~ rewrites/second)

result Code:

bit[1 .. 3] := false ;

foreach new state s do {

// "first update the bits in a consistent order"

bit’[1] := ’b(s) ;

bit’[2] := ’f(s) \/ ! ’e(s) /\ bit[2] ;

bit’[3] := ’d(s) /\ bit’[2] \/ ’c(s) /\ bit[3] ;

// "then check whether the formula is violated"

if ’a(s) /\ ! (bit[1] /\ bit’[3]) then

Error ;

// "finally, update the state of the monitor"

bit[1 .. 3] := bit’[1 .. 3]

}

Maude>

The K Module One should upload the next module whenever one wants to
use the K technique to define a language, logic or tool. Note that the module
below has nothing to do with our particular logic under consideration in this
paper; that is the reason for which we exiled it here.

fmod K is

sorts KConfigurationItem KConfiguration .

subsort KConfigurationItem < KConfiguration .

op empty : -> KConfiguration .

op __ : KConfiguration KConfiguration -> KConfiguration [assoc comm id: empty] .

sorts KComputationItem KNeComputation KComputation .

subsort KComputationItem < KNeComputation < KComputation .

op nil : -> KComputation .

op _->_ : KComputation KComputation -> KComputation [assoc id: nil] .

op _->_ : KNeComputation KNeComputation -> KNeComputation [ditto] .

sort KComputationList .

subsort KComputation < KComputationList .

op nil : -> KComputationList .

op _,_ : KComputationList KComputationList -> KComputationList [assoc id: nil] .

8



sort KResult KResultList .

subsorts KResult < KResultList < KComputation .

op nil : -> KResultList .

op _,_ : KResultList KResultList -> KResultList [assoc id: nil] .

op [_] : KComputationList -> KComputationItem .

op [_] : KResultList -> KComputationItem .

op [_|_] : KComputationList KResultList -> KComputationItem .

var K : KNeComputation . var Kl : KComputationList .

var R : KResult . var Rl : KResultList .

eq [K,Kl] = K -> [Kl | nil] .

eq R -> [K,Kl | Rl] = K -> [Kl | Rl,R] .

eq R -> [nil | Rl] = [Rl,R] .

endfm

To use K, after importing the module above, one should define one’s own con-
structors for configuration items (sort KConfigurationItem), for computation
items (sort KComputationItem), and for results (sort KResult). For our example,
we defined all these at the beginning of the module MONITOR-GENERATION.

